

# Motion of a Spinning Top

Jan Vrbik

Both approximate and exact solutions for the motion of a spinning top are constructed with the help of quaternions.

## ■ Quaternion Algebra

A quaternion is a four-dimensional quantity consisting of a scalar, say  $A$ , and a three-dimensional vector  $\mathbf{a}$ , collectively denoted  $\mathbb{A} \equiv (A, \mathbf{a})$ . Addition of two quaternions is component-wise,

$$(A, \mathbf{a}) \oplus (B, \mathbf{b}) = (A + B, \mathbf{a} + \mathbf{b}), \quad (1)$$

(we do not need to add quaternions in this article). Their multiplication follows the rule

$$(A, \mathbf{a}) \otimes (B, \mathbf{b}) = (AB - \mathbf{a} \cdot \mathbf{b}, A\mathbf{b} + B\mathbf{a} - \mathbf{a} \times \mathbf{b}). \quad (2)$$

It is important to note that such multiplication is *associative* (even though noncommutative). This can be verified by the following.

```
{A_, a_} ◦ {B_, b_} :=
  {A B - a.b, A b + B a - a × b} // TrigReduce // Simplify

({{A, {a1, a2, a3}} ◦ {B, {b1, b2, b3}}) ◦ {C, {c1, c2, c3}} -
  {A, {a1, a2, a3}} ◦ ({B, {b1, b2, b3}} ◦ {C, {c1, c2, c3}}) //
  Expand

{0, {0, 0, 0}}
```

When the scalar part of a quaternion is zero, it is called a pure quaternion. Since this constitutes an important special case, we will extend our definition of multiplication accordingly.

```
{A_, a_} ◦ b_ := {-a.b, A b - a × b} // TrigReduce // Simplify
```

The operation of conjugation simply changes the sign of the vector part of a quaternion.

```
 $\overline{\{A_, a_\}}$  := {A, -a}
```

Based on the corresponding Taylor expansion, it is possible to evaluate various functions of a quaternion. Of these, the most important is the exponential function, which, in this article, we need only with pure-quaternion arguments.

$$\mathbf{PQExp}[\mathbf{a\_}] := \mathbf{Module}\left[\left\{\mathbf{n} = \sqrt{\mathbf{a}.\mathbf{a}} \text{ // } \mathbf{PowerExpand}\right\}, \left\{\mathbf{Cos}[\mathbf{n}], \mathbf{a} / \mathbf{n} \mathbf{Sin}[\mathbf{n}]\right\}\right]$$

## ■ Rotation

Rotating a 3D vector  $\mathbf{x}$  with respect to an axis through the origin (with a unit direction of  $\hat{\mathbf{a}}$ ) by an angle  $a$  can be achieved by

$$\begin{aligned} \mathbf{x}' &= e^{-\alpha/2} \circ \mathbf{x} \circ e^{\alpha/2} = \left(\cos \frac{a}{2}, -\hat{\mathbf{a}} \sin \frac{a}{2}\right) \circ (\mathbf{x}_{\parallel} \oplus \mathbf{x}_{\perp}) \circ \left(\cos \frac{a}{2}, \hat{\mathbf{a}} \sin \frac{a}{2}\right) = \\ &\mathbf{x}_{\parallel} \oplus \mathbf{x}_{\perp} \circ (\cos a, \hat{\mathbf{a}} \sin a) = \mathbf{x}_{\parallel} \oplus (\mathbf{x}_{\perp} \cos a + \alpha \times \mathbf{x}_{\perp} \sin a), \end{aligned} \quad (3)$$

where  $\alpha = a \hat{\mathbf{a}}$  (a vector of length  $a$  and unit direction  $\hat{\mathbf{a}}$ ), and  $\mathbf{x}_{\parallel}$  and  $\mathbf{x}_{\perp}$  are the parallel and perpendicular (to  $\alpha$ ) parts of  $\mathbf{x}$ , respectively [1]. Note that  $\mathbf{x}_{\parallel}$  and  $\alpha$  commute, while  $\mathbf{x}_{\perp}$  and  $\alpha$  anticommute.

One can show that a quaternion  $\mathbb{R}$  has the form of  $e^{\alpha/2}$  if and only if  $\overline{\mathbb{R}} \circ \mathbb{R} \equiv \mathbb{R} \circ \overline{\mathbb{R}} = 1$  (a pure scalar). Alternately, the same  $\mathbb{R}$  can be parametrized by the three Euler angles (see the next section).

By taking  $\alpha = t \omega$ , where  $t$  is time, we achieve constant rotation with an uniform angular velocity  $\omega$ . Note that

$$\omega = 2 \overline{\mathbb{R}} \circ \dot{\mathbb{R}}, \quad (4)$$

where  $\mathbb{R} = e^{(\omega/2)t}$ , and  $\dot{\mathbb{R}}$  denotes its time derivative.

The previous formula can be generalized to any time-dependent  $\mathbb{R}$ , as can be seen by differentiating  $\mathbf{x}' = \overline{\mathbb{R}} \circ \mathbf{x} \circ \mathbb{R}$ , thus

$$\begin{aligned} \dot{\mathbf{x}}' &= \overline{\mathbb{R}} \circ \mathbf{x} \circ \dot{\mathbb{R}} + \dot{\overline{\mathbb{R}}} \circ \mathbf{x} \circ \mathbb{R} = \overline{\mathbb{R}} \circ \mathbf{x} \circ \mathbb{R} \circ \overline{\mathbb{R}} \circ \dot{\mathbb{R}} + \dot{\overline{\mathbb{R}}} \circ \mathbb{R} \circ \overline{\mathbb{R}} \circ \mathbf{x} \circ \mathbb{R} = \\ &\frac{\overline{\mathbb{R}} \circ \mathbf{x} \circ \mathbb{R} \circ \omega + \overline{\omega} \circ \overline{\mathbb{R}} \circ \mathbf{x} \circ \mathbb{R}}{2} = \frac{\mathbf{x}' \circ \omega - \omega \circ \mathbf{x}'}{2} = \omega \times \mathbf{x}', \end{aligned} \quad (5)$$

where  $\omega$  is now the instantaneous angular velocity at time  $t$ , which may change with time. Note that  $\overline{\mathbb{R}} \circ \dot{\mathbb{R}}$  must be a pure quaternion (take the time derivative of  $\overline{\mathbb{R}} \circ \mathbb{R} = 1$  to see that). Now, any rotation  $\mathbb{R}$  can be applied to the axes of the original (inertial) system of coordinates, thus creating the corresponding new, rotating (noninertial) frame. Components of any vector  $\mathbf{x}$  can then be expressed in either the old, or (often and more conveniently) the new coordinates; the latter will be given by  $\mathbb{R} \circ \mathbf{x} \circ \overline{\mathbb{R}}$ . The angular velocity  $\omega$ , in the new coordinates, will thus have components given by  $2 \dot{\mathbb{R}} \circ \overline{\mathbb{R}}$ . This becomes an important

and helpful tool: even though the laws of physics are normally valid only in inertial coordinates, the resulting equations often simplify when expressed in the rotating frame.

## ■ Spinning Top

Let us consider an axially symmetrical body (a “top”) of mass  $M$ , and parallel and perpendicular (with respect to the body’s symmetry axis) moments of inertia equal to  $J_{\parallel}$  and  $J_{\perp}$ , respectively. We make it spin around its axis, place the bottom tip of its (usually tilted) axis on a nonslip horizontal plane (a desk), and let it continue its motion subject to the vertical gravity field. Let us also assume that the top’s center of mass is at a distance  $r$  from its point of contact, that the gravitational acceleration is  $g$ , and that there is no friction.

To investigate the top’s motion, we introduce an inertial coordinate system with its origin at the point of contact and its  $z$  direction pointing vertically upward. First we position a motionless top vertically (its symmetry axis aligned with the  $z$  direction) on the desk’s surface. We then rotate it using a fully general, time-dependent rotation  $\mathbb{R}$ , parametrized by three Euler angles, thus

$$\mathbb{R} = e^{\mathbb{k} \phi/2} \circ e^{\mathbb{i} \theta/2} \circ e^{\mathbb{j} \psi/2}, \quad (6)$$

where  $\phi$  is a rapidly increasing function of time that represents the actual spinning of the top around its symmetry axis (this rotation is applied first), followed by tilting the spinning top by an angle  $\theta$  (potentially a slowly varying function of time) around the inertial  $x$  direction, further followed by slowly rotating the spinning, tilted top around the inertial  $z$  direction by an angle  $\psi$ , thus creating a so-called precession. Also:  $\mathbb{i}$ ,  $\mathbb{j}$  (not used in the last formula), and  $\mathbb{k}$  are unit vectors along the  $x$ ,  $y$ , and  $z$  directions, respectively, in their pure-quaternion form.

## □ Equations of Motion

We now use the Lagrange technique to find the corresponding equations for the three Euler angles. The kinetic energy of the top equals one-half of the scalar product of its angular velocity with its angular momentum, as its motion is purely rotational. In the rotating frame (this is when it becomes handy), this equals

$$\frac{J_{\parallel} \omega_z^2 + (J_{\perp} + M r^2) (\omega_x^2 + \omega_y^2)}{2}, \quad (7)$$

where  $\omega_x$ ,  $\omega_y$ ,  $\omega_z$  are the rotating-frame coordinates of  $\omega$ , and  $J_{\perp} + M r^2$  (denoted  $J$  from now on; similarly,  $J_{\parallel}$  will be called  $J_3$ ) is the top’s moment of inertia with respect to a line perpendicular to its symmetry axis, and passing through the origin (which is displaced by  $r$  from its center of mass). Usually  $J > J_3$  (a thin circular disk needs  $r$  to be bigger than only one-half of its radius to meet this condition), but our results are valid when  $J \leq J_3$  as well.

Subtracting the potential energy  $M g r \cos \theta \equiv J m \cos \theta$  (by the definition of  $m$ ) yields the resulting Lagrangian.

$$\begin{aligned} \mathbf{i} &= \{1, 0, 0\}; \mathbf{j} = \{0, 1, 0\}; \mathbf{k} = \{0, 0, 1\}; \\ \mathbf{R} &= (\text{PQExp}[\mathbf{k} / 2 \phi[t]] \circ \text{PQExp}[\mathbf{i} / 2 \theta[t]]) \circ \text{PQExp}[\mathbf{k} / 2 \psi[t]]; \\ \omega &= (2 \mathbf{D}[\mathbf{R}, t] \circ \overline{\mathbf{R}})[[2]]; \\ \mathbf{L} &= \{\mathbf{J}, \mathbf{J}, \mathbf{J}_3\} \cdot \omega^2 / 2 - \mathbf{J} m \text{Cos}[\theta[t]] // \text{Simplify} \\ &= \frac{1}{2} \left( -2 \mathbf{J} m \text{Cos}[\theta[t]] + \mathbf{J} \theta'[t]^2 + \right. \\ &\quad \left. \mathbf{J} \text{Sin}[\theta[t]]^2 \psi'[t]^2 + \mathbf{J}_3 (\phi'[t] + \text{Cos}[\theta[t]] \psi'[t])^2 \right) \end{aligned}$$

Since the result is free of  $\phi$  and  $\psi$  (it contains only their time derivatives), the corresponding  $\frac{\partial L}{\partial \phi}$  and  $\frac{\partial L}{\partial \psi}$  must be constants of motion [2]. The remaining equation is then obtained from

$$\frac{d}{d\theta} \left( \frac{\partial L}{\partial \dot{\theta}} \right) = \frac{\partial L}{\partial \theta}. \quad (8)$$

This yields the following.

$$\begin{aligned} \mathbf{cm}_1 &= \mathbf{D}[\mathbf{L}, \phi'[t]] // \text{Simplify} \\ \mathbf{cm}_2 &= \mathbf{D}[\mathbf{L}, \psi'[t]] // \text{Simplify} \\ \mathbf{eq} &= \mathbf{D}[\mathbf{D}[\mathbf{L}, \theta'[t]], t] - \mathbf{D}[\mathbf{L}, \theta[t]] // \text{Simplify} \\ &= \mathbf{J}_3 (\phi'[t] + \text{Cos}[\theta[t]] \psi'[t]) \\ &= \mathbf{J} \text{Sin}[\theta[t]]^2 \psi'[t] + \text{Cos}[\theta[t]] \mathbf{J}_3 (\phi'[t] + \text{Cos}[\theta[t]] \psi'[t]) \\ &= -\text{Sin}[\theta[t]] \left( -\mathbf{J}_3 \psi'[t] (\phi'[t] + \text{Cos}[\theta[t]] \psi'[t]) + \right. \\ &\quad \left. \mathbf{J} (m + \text{Cos}[\theta[t]] \psi'[t]^2) \right) + \mathbf{J} \theta''[t] \end{aligned}$$

The first two expressions are constants of motion, while the last one must be equal to zero.

Introducing a new variable  $\Phi$  by

$$\dot{\phi} = \frac{J \dot{\Phi} + (J - J_3) \dot{\psi} \cos \theta}{J_3}, \quad (9)$$

the last three expressions simplify further. It is also important to realize that the transformation (9) replaces only the  $\phi$  parameter (the speed of spinning); if we are interested in the behavior of  $\psi$  and  $\theta$  only (as is usually the case), we get the same solution regardless of whether we use  $\phi$  (in the old set of equations) or  $\Phi$  (in the new one).

So, let us see what the new results look like.

```

cm1 = cm1 / J /. ϕ' [t] → (J ϖ' [t] + (J - J3) ψ' [t] Cos[θ[t]]) / J3 //
Simplify
cm2 = cm2 / J /. ϕ' [t] → (J ϖ' [t] + (J - J3) ψ' [t] Cos[θ[t]]) / J3 //
Simplify
eq = eq / J /. ϕ' [t] → (J ϖ' [t] + (J - J3) ψ' [t] Cos[θ[t]]) / J3 //
Simplify

```

$$\varpi'[t] + \text{Cos}[\theta[t]] \psi'[t]$$

$$\text{Cos}[\theta[t]] \varpi'[t] + \psi'[t]$$

$$-m \text{Sin}[\theta[t]] + \text{Sin}[\theta[t]] \varpi'[t] \psi'[t] + \vartheta''[t]$$

The equations are now much simpler, and the number of parameters has been reduced from three to only one (namely  $m$ , whose dimension is  $\text{sec}^{-2}$ ).

We can now easily infer a possibility of what we call a “steady” solution, with a constant value of  $\theta$  (say  $\theta_0$ ) and

$$\begin{aligned} \Phi &= \dot{\Phi}_0 t + \Phi_0, \\ \psi &= \dot{\psi}_0 t + \psi_0, \end{aligned} \tag{10}$$

where  $\dot{\Phi}_0$  and  $\dot{\psi}_0$  are two constants whose product must be equal to  $m$ . To relate the precession speed  $\dot{\psi}_0$  to the actual spinning speed, say  $\dot{\phi}_0$ , we have to solve (based on (9))

$$J_3 \dot{\phi}_0 = \frac{m J}{\dot{\psi}_0} + (J - J_3) \dot{\psi}_0 \cos \theta_0 \tag{11}$$

for  $\dot{\psi}_0$ , which yields

$$\dot{\psi}_0 = \frac{J_3 \dot{\phi}_0 \pm \sqrt{(J_3 \dot{\phi}_0)^2 - 4 m J (J - J_3) \cos \theta_0}}{2 (J - J_3) \cos \theta_0}, \tag{12}$$

where the negative sign corresponds to the usually observed “slow” precession (assuming, from now on, that  $\dot{\phi}_0 > 0$ ), whereas the positive sign yields a hard-to-achieve “fast” precession. When  $J = J_3$  (spherically symmetrical top), we get only the “slow” solution  $\dot{\psi}_0 = \frac{m}{\dot{\phi}_0}$ .

For a fast-spinning top, the previous formula yields, to a good approximation,  $\dot{\psi}_0 = \frac{J m}{J_3 \dot{\phi}_0}$ ; the precession speed is practically independent of  $\theta_0$ .

## □ Approximate Solution

The actual motion is usually more complicated than a steady solution of the previous section, as  $\theta$  may undergo periodic oscillations, called nutation. Assuming that the amplitude of these is small, we can expand the three dependent variables of our equations around the steady-state solution:

$$\begin{aligned}\theta &= \theta_s + \epsilon \theta_1, \\ \dot{\psi} &= \dot{\psi}_s + \epsilon \dot{\psi}_1, \\ \dot{\Phi} &= \frac{m}{\dot{\psi}_s} + \epsilon \dot{\Phi}_1,\end{aligned}\tag{13}$$

where the first term of each right-hand side is a constant, the second term remains time dependent, and  $\epsilon$  is small. We substitute these into the three equations, neglecting  $\epsilon^2$  and higher powers of  $\epsilon$ .

$$\begin{aligned}\mathbf{aux} &= \{\mathbf{cm}_1, \mathbf{cm}_2, \mathbf{eq}\} /. \left\{ \theta[t] \rightarrow \theta_s + \epsilon \theta_1[t], \theta'[t] \rightarrow \epsilon \theta_1'[t], \right. \\ &\quad \left. \psi'[t] \rightarrow \dot{\psi}_s + \epsilon \dot{\psi}_1'[t], \Phi'[t] \rightarrow m / \dot{\psi}_s + \epsilon \Phi_1'[t] \right\}; \\ \mathbf{approx} &= \mathbf{Coefficient}[\mathbf{Series}[\mathbf{aux}, \{\epsilon, 0, 1\}], \epsilon] \\ &\left\{ -\sin[\theta_s] \dot{\psi}_s \theta_1[t] + \Phi_1'[t] + \cos[\theta_s] \dot{\psi}_1'[t], \right. \\ &\quad \left. - \frac{m \sin[\theta_s] \theta_1[t]}{\dot{\psi}_s} + \cos[\theta_s] \Phi_1'[t] + \dot{\psi}_1'[t], \right. \\ &\quad \left. \sin[\theta_s] \dot{\psi}_s \Phi_1'[t] + \frac{m \sin[\theta_s] \dot{\psi}_1'[t]}{\dot{\psi}_s} + \theta_1''[t] \right\}\end{aligned}$$

Each of the three expressions must be equal to zero.

We solve the first two equations for  $\psi_1'$  and  $\Phi_1'$  and substitute into the last equation.

```
aux = Solve[{approx[[1]] == 0, approx[[2]] == 0},  
  {psi_1'[t], phi_1'[t]}] // Simplify  
approx[[3]] /. aux // Simplify
```

$$\left\{ \left\{ \psi_1'[t] \rightarrow \frac{\csc[\theta_s] \left( m - \cos[\theta_s] \dot{\psi}_s^2 \right) \theta_1[t]}{\dot{\psi}_s}, \right. \right.$$

$$\left. \left. \Phi_1'[t] \rightarrow \frac{\csc[\theta_s] \left( -m \cos[\theta_s] + \dot{\psi}_s^2 \right) \theta_1[t]}{\dot{\psi}_s} \right\} \right\}$$

$$\left\{ \left( -2 m \cos[\theta_s] + \frac{m^2}{\dot{\psi}_s^2} + \dot{\psi}_s^2 \right) \theta_1[t] + \theta_1''[t] \right\}$$

Solving for  $\theta_1$  (and, subsequently, for  $\psi_1$  and  $\Phi_1$ ) is now quite easy. Note that the coefficient of  $\theta_1$  in the last expression is always non-negative.

Also note that  $\dot{\psi}$  will be always positive (assuming “slow” precession), whenever the following condition is met:

$$\frac{(m - \cos \theta_s \dot{\psi}_s^2) \epsilon}{\dot{\psi}_s \sin \theta_s} < \dot{\psi}_s. \quad (14)$$

Reversing the above inequality results in so-called looping orbits (in terms of the path of the axis, when displayed on a unit sphere: see below), with  $\dot{\psi}$  periodically changing direction; making the two sides of (14) equal to each other results in cuspidal orbits (when  $\theta$  reaches its smallest value,  $\dot{\psi}$  becomes zero).

If desired, one can extend the approximate solution to achieve  $\epsilon^2$  (and higher) accuracy.

## □ Exact Solution

One can show that the original differential equation for  $\theta$ , namely

$$\ddot{\theta} + \dot{\Phi} \dot{\psi} \sin \theta - m \sin \theta = 0 \quad (15)$$

has the following constant of motion.

$$\mathbf{cm}_3 = \frac{1}{2} \left( \theta'[t]^2 + \psi'[t]^2 + \Phi'[t]^2 \right) + \psi'[t] \Phi'[t] \cos[\theta[t]] + m \cos[\theta[t]];$$

This can be easily verified.

```

D[cm3, t] / θ'[t] /.
Solve[{D[cm1, t] == 0, D[cm2, t] == 0}, {ψ''[t], ϕ''[t]}][[
1]] // Simplify

-m Sin[θ[t]] + Sin[θ[t]] ϕ'[t] ψ'[t] + θ''[t]

```

Making  $cm_1$  and  $cm_2$  equal to their initial values, we can solve for  $\dot{\psi}$  and  $\dot{\phi}$  and substitute these into  $cm_3$  (minus its initial value, so that the resulting expression must be equal to zero).

```

cm3 = cm3 - 1/2 (θ̇02 + ψ̇02 + ϕ̇02) - ψ̇0 ϕ̇0 Cos[θ0] - m Cos[θ0];
eq =
cm3 /.
(aux = Solve[{cm1 == ϕ̇0 + ψ̇0 Cos[θ0], cm2 == ϕ̇0 Cos[θ0] + ψ̇0},
{ϕ'[t], ψ'[t]}][[1]]) // Simplify;

```

The resulting differential equation can be simplified by the  $\theta = 2 \arccos \lambda$  transformation [3].

```

eqn = TrigExpand[eq /. θ0 → 2 q] /. {θ̇0 → 0, q → ArcCos[λ0]} //
Simplify;
eqn = TrigExpand[eqn /. θ[t] → 2 q] /.
{θ'[t] → -2 λ'[t] / √(1 - λ[t]2), q → ArcCos[λ[t]]} //
Simplify;
eqn = Collect[eqn (-1 + λ[t]2) / 2 // Simplify, {λ'[t], λ[t]},
Simplify]

```

$$\begin{aligned}
& \frac{1}{2} \lambda_0^2 \left( 2m + \dot{\phi}_0^2 + 2\lambda_0^2 \dot{\phi}_0 \dot{\psi}_0 + \dot{\psi}_0^2 \right) - \frac{\lambda_0^4 (\dot{\phi}_0 + \dot{\psi}_0)^2}{4\lambda[t]^2} + \\
& \frac{1}{4} \left( -4m - \dot{\phi}_0^2 + 2\dot{\phi}_0 \dot{\psi}_0 - \dot{\psi}_0^2 - 4\lambda_0^2 (m + \dot{\phi}_0 \dot{\psi}_0) \right) \lambda[t]^2 + m\lambda[t]^4 - \lambda'[t]^2
\end{aligned}$$



To simplify the subsequent solution, we have assumed (without loss of generality) that the initial time has been chosen to coincide with  $\theta$  reaching its maximum value (and, consequently,  $\dot{\lambda}_0 = 0$ ). The above equation can then be solved, using specific initial conditions, which must satisfy  $\dot{\Phi}_0 \dot{\psi}_0 \geq m$  (due to (15), since  $\ddot{\theta}_0 \leq 0$ ); here  $\lambda_{sq}$  provides the solution for  $\lambda^2$ .


```
{eqcf,  $\Lambda$ , aux} =
  {Coefficient[eqn +  $\lambda'$ [t]^2,  $\lambda$ [t], {4, 2, 0, -2}],  $\lambda_0^2$ , aux} /.
  Thread[{ $\lambda_0$ ,  $\dot{\psi}_0$ ,  $\ddot{\theta}_0$ , m} -> {0.85, 1, 3, 1}];
 $\delta = x /. \text{Solve}[0 == \text{eqcf}.\{x^2 + 3 \Lambda x + 3 \Lambda^2, x + 2 \Lambda, 1, 0\}, x][[1]]$ ;
 $\mu = \sqrt{-\text{eqcf}.\{3 \Lambda + \delta, 1, 0, 0\}}$ ;
 $k = \delta \text{eqcf}[[1]] / \mu^2$ ;
 $\lambda_{sq} = \Lambda + \delta \text{JacobiSN}[\mu t, k]^2$ 

0.7225 + 0.156255 JacobiSN[1.60195 t, 0.0608884]^2
```

This can be easily transformed back to  $\theta$ ; one can then find the corresponding  $\Phi$  and  $\psi$  by analytically integrating the expressions for  $\dot{\psi}$  and (if desired)  $\dot{\Phi}$ . Finally, the results can be displayed graphically.

```
 $\dot{\psi}_{sol} =$ 
   $\psi'[t] /. \text{aux} /. \{\text{Cos}[\theta[t]] -> 2 \lambda[t]^2 - 1, \text{Cos}[\theta_0] -> 2 \Lambda - 1\} /.$ 
   $\{\lambda[t]^2 \rightarrow \Lambda + \delta J, \lambda[t]^4 \rightarrow (\Lambda + \delta J)^2\}$ ;
 $\psi_{sol} = 3.7 + \text{Apart}[\dot{\psi}_{sol}, J] /.$ 
   $\left\{ \frac{c_-}{a_- + b_- J} \rightarrow \right.$ 
   $\left( c \text{EllipticPi}\left[-\frac{b}{a}, \text{JacobiAmplitude}[\mu t, k], k\right] \right.$ 
   $\left. \left. \text{JacobiDN}[\mu t, k] \right) / \left( a \mu \sqrt{1 - k \text{JacobiSN}[\mu t, k]^2} \right) \right\}$ ;

 $\theta_{sol} = 2 \text{ArcCos}\left[\sqrt{\lambda_{sq}}\right]$ ;
Show[ParametricPlot3D[
  {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]}, {v, 0, Pi},
  {u, 0, 2 Pi}],
ParametricPlot3D[
  Evaluate[{Cos[ $\psi_{sol}$ ] Sin[ $\theta_{sol}$ ], Sin[ $\psi_{sol}$ ] Sin[ $\theta_{sol}$ ],
    Cos[ $\theta_{sol}$ ]}], {t, 0, 10}]]
```



One can now compute and display the top's motion under various initial conditions, check the accuracy of the approximate solution, and so forth. We will leave it for you to explore. Have fun!

## ■ References

- [1] D. Hestenes, *New Foundations for Classical Mechanics*, 2nd ed., Dordrecht/Boston/London: Kluwer Academic Publishers, 1999.
- [2] H. Goldstein, *Classical Mechanics*, 2nd ed., Reading, MA: Addison-Wesley, 1980.
- [3] E. T. Whittaker, *A Treatise on the Analytical Dynamics of Particles and Rigid Bodies*, 2nd ed., Cambridge: Cambridge University Press, 1917.  
[www.archive.org/details/atreatiseonanal00whitgoog](http://www.archive.org/details/atreatiseonanal00whitgoog).

J. Vrbik, "Motion of a Spinning Top," *The Mathematica Journal*, 2012. [dx.doi.org/doi:10.3888/tmj.14-4](https://doi.org/10.3888/tmj.14-4).

## About the Author

### **Jan Vrbik**

*Department of Mathematics, Brock University*  
*500 Glenridge Ave., St. Catharines*  
*Ontario, Canada, L2S 3A1*  
[jvr bik@brocku.ca](mailto:jvr bik@brocku.ca)